


기본 문법

IDL 사용자들에게는 매우 친근한 문법 그대로입니다. 다음을 입력해 보세요.

```
IDL> data=randomu(-1L, 10)
IDL> print, data
      0.41 0.09 0.75 0.52 0.93 0.38
      0.65 0.06 0.72 0.67
IDL> bp=barplot(data)
```

인수를 하나(위 예에서는 data)만 넘겨 주면 X 축은 자동으로 0부터 시작하는 인덱스가 됩니다. X 값도 물론 지정해 줄 수 있습니다. 예를 들어, 0부터 시작하지 않고 1부터 시작하는 X축을 그리고 싶다면 다음과 같이 합니다.

```
IDL> x=indgen(10)+1
IDL> print, x
      1 2 3 4 5 6 7 8 9 10
IDL> bp2=barplot(x, data)
```




barplot() 함수는 IDL 사용자들에게는 너무 당연하게도 다음과 같은 기본 문법을 가지고 있습니다.

1. variable = BARPLOT(값)
2. variable = BARPLOT(위치, 값)

색 지정

BARPLOT()에서 COLOR는 테두리 색을, FILL_COLOR는 내부를 칠하는 색을 지정하는 키워드입니다.




```
IDL> bp3=barplot(data, FILL_COLOR='yellow', $
      COLOR='red')
```

그래데이션 색 지정


BOTTOM_COLOR 키워드를 사용하면, 그래프의 맨 아래쪽 색을 지정할 수 있습니다. 이 키워드를 사용하면, BOTTOM_COLOR로 시작하여 FILL_COLOR로 끝나는 단계적인 색 변화를 줄 수 있습니다.

```
IDL> bp4=barplot(data, BOTTOM_COLOR='white', $
      FILL_COLOR='violet', color='tomato')
```




그래데이션의 기본 설정은 각 막대마다 가장 낮은 값이 BOTTOM_COLOR, 가장 높은 값이 FILL_COLOR가 되도록 순차적으로 색을 배정합니다. 앞의 그래프에서 보면 같은 값인데도(예를 들어 0.5) 다른 농도로 표출되어 있습니다. 만일, 이 것이 문제가 된다면(같은 값은 같은 색으로 표출되게 하고 싶다면) C_RANGE 키워드를 사용하면 됩니다. C_RANGE 범위보다 작은 값은 모두 BOTTOM_COLOR로, 범위보다 큰 값은 모두 FILL_COLOR로 채워집니다. 범위 안쪽은 단계적으로 변화합니다.

```
bp5=barplot(data, BOTTOM_COLOR='INDIGO', $
      FILL_COLOR='LIME', C_RANGE=[0, 0.7])
```



다중 막대 그래프

막대 그래프의 특기는 '비교'입니다. X축에 따른 값의 비교도 중요하지만, 같은 X축을 가지는 값들끼리의 비교가 중요할 때도 있습니다. 예를 들어서, 2008년, 2009년, 2010년의 월 평균 기온을 표출하고자 한다면, X축이 월로 설정이 되어야 겠지만, 연도별 비교도 표출이 되어야 합니다. 다중 막대 그래프를 이용하여 이를 표출하면 다음과 같습니다.



```
t08=[1.1,2.7,7.1,13.9,16.3,21.5,26.7,25.6, $
      22.7,17.3,9.0,4.1]
t09=[1.2,5.3,7.4,12.0,18.0,21.9,24.5,25.0, $
      21.6,16.2,9.6,6.1]
t10=[1.3,4.0,6.2,9.1,16.0,20.7,24.4,25.1, $
      21.5,10.9,8.6,3.9]
month=indgen(12)+1 ;1~12

b08=barplot(month, t08, NBARS=3, INDEX=0, $
  BOTTOM_COLOR='white', FILL_COLOR='GOLD', $
  NAME='2008', TITLE='Monthly Temperature', $
  XTITLE='Month', YTITLE='Temperature')

b09=barplot(month, t09, NBARS=3, INDEX=1, $
  BOTTOM_COLOR='white', FILL_COLOR='RED', $
  NAME='2009', /OVERPLOT)

b10=barplot(month, t10, NBARS=3, INDEX=2, $
  BOTTOM_COLOR='white', FILL_COLOR='NAVY', $
  NAME='2010', /OVERPLOT)


l=legend(target=[b08, b09, b10], $
  position=[1, 25], /DATA)
```

다중 막대그래프에서 중요한 문법은 NBARS 키워드, INDEX 키워드, OVERPLOT 키워드입니다. NBARS 키워드를 이용하여 전체 몇 개의 계열을 그릴 것인지 설정하고, INDEX 키워드는 그 중 몇 번째 계열의 데이터를 그리는지 지정합니다. /OVERPLOT 키워드를 이용하여 앞의 그래프에 중첩하여 그린다는 지정도 빼 놓지 마세요.

막대그래프 쌓기

계열 간의 비교라면 다중 막대 그래프로 표출하는 것이 효과적이지만, 계열간의 합산을 표출하는 경우라면, 막대그래프 쌓기가 좋을 것입니다.

예를 들어, 연도별 발전설비 용량을 막대그래프로 표출하되, 에너지원 별로 구분을 하고자 하는 경우 다음과 같이 그래프를 생성할 수 있습니다.



```
year=indgen(10)+2000 ;2000~2009
pnuclear=[13.7, 13.7, 15.7, 15.7, 16.7, $
           17.7, 17.7, 17.7, 17.7, 17.7]
pcoal=[14, 15.5, 15.9, 15.9, 17.5, 17.9, $
        18.5, 20.6, 23.7, 24.2]
pgas=[12.7, 12.9, 13.6, 14.5, 15.7, 16.4, $
       17.4, 18.5, 18.5, 18.4]

bn=barplot(year, pnuclear, name='Nuclear', $
  Title='Electric Gen. Equip.', xtitle='Year', $
  ytitle='Equip. Capacity.(GW)', $
  fill_color='crimson')

bc=barplot(year, pnuclear+pcoal, $
  BOTTOM_VALUES=pnuclear, name='Coal', $
  fill_color='dark_gray', /OVERPLOT)

bg=barplot(year, pnuclear+pcoal+pgas, $
  BOTTOM_VALUES=pnuclear+pcoal, name='Gas', $
  fill_color='dark_orange', /OVERPLOT)


l=legend(target=[bn, bc, bg], $
  position=[1999, 58], /DATA)
```

여기에서도 /OVERPLOT 키워드가 사용됩니다. 중요한 키워드는 BOTTOM_VALUES인데, 막대그래프의 시작점을 지정합니다.

가로 방향의 막대 그래프

/HORIZONTAL 키워드를 사용하면 막대그래프가 가로 방향으로 생성됩니다. 그 밖의 모든 사용법은 똑같습니다.

다만, X, Y 좌표계가 바뀌므로, XTITLE과 YTITLE을 제대로 지정해 주었는지는 확인하십시오.



```
IDL> bh=barplot(month, t08, /HORIZONTAL, $
  xtitle='Temperature', ytitle='Month')
```